

MANUAL DE ESTUDIO DEL ALUMNO

EXCEL INTERMEDIO – AVANZADO

ENCUENTRO 1

Funciones lógicas, análisis y búsqueda de información

Caso práctico: Laboratorios BIOVAC

Material didáctico para formación presencial

Duración del encuentro: 4 horas

Contenido

- 1. Presentación del encuentro
- 2. Caso de estudio: Laboratorios BIOVAC
- 3. Objetivos de aprendizaje
- 4. Metodología de trabajo
- 5. Conceptos fundamentales antes de comenzar
- 6. Función SI
- 7. SI anidado
- 8. Funciones Y y O
- 9. CONTAR.SI
- 10. CONTAR.SI.CONJUNTO
- 11. SUMAR.SI
- 12. SUMAR.SI.CONJUNTO
- 13. Funciones de búsqueda y referencia
- 14. BUSCARV
- 15. BUSCARH
- 16. ÍNDICE
- 17. COINCIDIR
- 18. ÍNDICE + COINCIDIR
- 19. Errores frecuentes y cómo resolverlos
- 20. Buenas prácticas de trabajo
- 21. Ejercicios guiados
- 22. Ejercicios de práctica individual
- 23. Reto integrador
- 24. Autoevaluación
- 25. Hoja de fórmulas de consulta rápida
- 26. Glosario

1. Presentación del encuentro

En este primer encuentro vamos a utilizar Excel como una herramienta de análisis y consulta, no solamente como una calculadora. El hilo conductor será Laboratorios BIOVAC, una empresa ficticia dedicada a la producción, almacenamiento y distribución de vacunas. Todos los datos utilizados en este manual son simulados y tienen únicamente fines educativos.

La meta es aprender a convertir una base de datos en información útil: clasificar registros, contar casos, sumar resultados bajo condiciones y recuperar información automáticamente a partir de códigos.

¿Qué aprenderás?

- Construir decisiones automáticas con SI, Y y O.
- Crear clasificaciones con SI anidado.
- Contar registros con uno o varios criterios.
- Sumar información con uno o varios criterios.
- Buscar datos vertical y horizontalmente.
- Comprender la lógica de ÍNDICE y COINCIDIR.
- Combinar funciones para resolver consultas más flexibles.
- Detectar y corregir errores frecuentes en fórmulas.
- Aplicar buenas prácticas para trabajar con bases de datos.

Duración sugerida

Bloque	Tiempo	Enfoque
Inicio y diagnóstico	15 min	Repaso y contexto
Funciones lógicas	45 min	SI, SI anidado, Y, O
Funciones de conteo	40 min	CONTAR.SI y CONTAR.SI.CONJUNTO
Funciones de suma	40 min	SUMAR.SI y SUMAR.SI.CONJUNTO
Descanso	10 min	—
Búsquedas	30 min	BUSCARV y BUSCARH
ÍNDICE y COINCIDIR	40 min	Búsqueda flexible
Reto y evaluación	20 min	Aplicación integral

2. Caso de estudio: Laboratorios BIOVAC

BIOVAC necesita mejorar el análisis de sus registros de producción. El personal administrativo recibe información de lotes y debe responder preguntas como: ¿cuántos lotes están aprobados?, ¿cuánto se produjo en una planta?, ¿qué responsable tiene un lote determinado? y ¿qué información corresponde a un código?

Base principal de trabajo

Lote	Fecha	Vacuna	Planta	Responsable	Unidades	Calidad	Estado
BV-001	03/08/2026	BIOVAC-20	Managua	Ana López	25000	Aprobado	Liberado
BV-002	04/08/2026	Managua	León	Carlos Pérez	18000	Pendiente	En revisión
BV-003	05/08/2026	BIOVAC-FLU	Managua	Ana López	30000	Aprobado	Liberado
BV-004	06/08/2026	BIOVAC-P	León	Pedro Ruiz	22000	Rechazado	No liberar
BV-005	07/08/2026	BIOVAC-20	Managua	María Soto	28000	Aprobado	Liberado

Nota: la tabla completa del curso se trabajará en el archivo Excel suministrado por el instructor.

Catálogo de vacunas

Código	Vacuna	Presentación	Costo unitario	Temperatura
VAC001	BIOVAC-20	1 dosis	450	2–8 °C
VAC002	BIOVAC-FLU	1 dosis	380	2–8 °C
VAC003	BIOVAC-P	10 dosis	1,250	2–8 °C
VAC004	BIOVAC-R	1 dosis	520	2–8 °C

3. Objetivos de aprendizaje

1. Construir fórmulas que permitan tomar decisiones automáticas.
2. Combinar condiciones mediante Y y O.
3. Crear clasificaciones mediante SI anidado.
4. Analizar una base mediante criterios de conteo.
5. Analizar una base mediante criterios de suma.
6. Utilizar BUSCARV y BUSCARH para recuperar información.
7. Comprender la relación entre ÍNDICE y COINCIDIR.
8. Aplicar búsquedas en un escenario administrativo.
9. Identificar errores comunes en fórmulas.
10. Resolver un caso integrador con varias funciones.

Criterio de logro

Se considera que el objetivo del encuentro se alcanzó cuando el participante puede recibir una base de datos similar a la de BIOVAC, interpretar una necesidad administrativa, seleccionar la función adecuada y construir una solución sin depender de copiar una fórmula sin comprenderla.

4. Metodología de trabajo

Cada tema seguirá cuatro pasos:

11. Entender el problema: ¿qué necesitamos obtener?
12. Identificar la lógica: ¿qué condición, rango o criterio necesitamos?
13. Construir la fórmula: escribir y comprobar la solución.
14. Interpretar el resultado: explicar qué significa para la empresa.

Regla de oro

Antes de escribir una fórmula, expresa el problema en lenguaje natural. Si puedes explicar qué debe hacer Excel, será mucho más fácil construir la fórmula correcta.

5. Conceptos fundamentales antes de comenzar

Referencias relativas y absolutas

Una referencia relativa cambia cuando copiamos una fórmula. Una referencia absoluta permanece fija. El símbolo \$ permite fijar columna, fila o ambas.

Referencia	Qué ocurre al copiar
A2	Fila y columna pueden cambiar
\$A\$2	No cambia
A\$2	La fila permanece fija
\$A2	La columna permanece fija

Separador de argumentos

En esta guía se utilizan fórmulas con punto y coma (;), configuración habitual de Excel en español. Dependiendo de la configuración regional, tu Excel puede mostrar coma (,).

Coincidencia exacta

En búsquedas de códigos, normalmente queremos coincidencia exacta. En BUSCARV y BUSCARH esto se indica con FALSO.

6. Función SI

La función SI permite tomar una decisión: si una condición se cumple, devuelve un resultado; si no se cumple, devuelve otro.

Sintaxis

=SI(prueba_lógica; valor_si_verdadero; valor_si_falso)

Ejemplo BIOVAC

Regla: si el control de calidad está aprobado, el lote puede liberarse; de lo contrario, queda en revisión.

=SI(G2="Aprobado";"Liberado";"En revisión")

Cómo pensar la fórmula

- ¿Qué estoy evaluando? → G2.
- ¿Qué debe cumplir? → que sea igual a Aprobado.
- ¿Qué pasa si cumple? → Liberado.
- ¿Qué pasa si no cumple? → En revisión.

Ejercicio guiado 1

En la columna Estado de liberación, crea una fórmula que muestre "Liberado" si Calidad es "Aprobado" y "No liberar" en cualquier otro caso.

Tu fórmula:

Resultado esperado: los lotes aprobados deben aparecer como Liberado.

7. SI anidado

Un SI anidado utiliza una función SI dentro de otra para evaluar varias categorías.

Ejemplo

Clasificar la producción según unidades:

Unidades	Clasificación
30,000 o más	Alta
20,000 a 29,999	Media
Menos de 20,000	Baja

=SI(F2>=30000;"Alta";SI(F2>=20000;"Media";"Baja"))

Orden de evaluación

Excel evalúa de izquierda a derecha. Por eso conviene colocar primero la condición más exigente y continuar hacia las siguientes.

Ejercicio guiado 2

Clasifica las unidades producidas con estas reglas:

- $\geq 40,000 \rightarrow$ Muy alta
- $\geq 30,000 \rightarrow$ Alta
- $\geq 20,000 \rightarrow$ Media
- $< 20,000 \rightarrow$ Baja

=SI(F2>=40000;"Muy alta";SI(F2>=30000;"Alta";SI(F2>=20000;"Media";"Baja")))

Escribe la fórmula sin mirar el ejemplo anterior y luego comprueba.

8. Funciones Y y O

Y

Y devuelve VERDADERO solamente cuando todas las condiciones se cumplen.

=Y(condición1; condición2; ...)

Ejemplo: un lote puede ser liberado automáticamente si la calidad está aprobada Y tiene al menos 20,000 unidades.

=SI(Y(G2="Aprobado";F2>=20000);"Liberar";"Revisar")

O

O devuelve VERDADERO cuando al menos una condición se cumple.

=O(condición1; condición2; ...)

Ejemplo: marcar un lote como prioritario si tiene menos de 20,000 unidades O si su calidad está pendiente.

=SI(O(F2<20000;G2="Pendiente");"Prioritario";"Normal")

Diferencia clave

Función	Lógica	Ejemplo
Y	Todas deben cumplirse	Aprobado Y $\geq 20,000$
O	Al menos una debe cumplirse	<20,000 O Pendiente

Ejercicio guiado 3

Un lote es "Listo para distribución" cuando Calidad = Aprobado Y Unidades $\geq 25,000$. Construye la fórmula.

9. CONTAR.SI

CONTAR.SI cuenta cuántas celdas cumplen un criterio.

=CONTAR.SI(rango; criterio)

Ejemplo

¿Cuántos lotes están aprobados?

=CONTAR.SI(G2:G101;"Aprobado")

El rango debe contener la información que vamos a revisar y el criterio indica qué queremos contar.

Ejemplos de criterios

Pregunta	Criterio
Lotes de Managua	"Managua"
Lotes aprobados	"Aprobado"
Producciones mayores a 25,000	>25000
Lotes distintos de aprobados	"<>Aprobado"

Ejercicio guiado 4

En la base del archivo Excel, calcula:

- Cantidad de lotes de la planta Managua.
- Cantidad de lotes aprobados.
- Cantidad de lotes con más de 25,000 unidades.

Resultados: _____ / _____ / _____

10. CONTAR.SI.CONJUNTO

CONTAR.SI.CONJUNTO cuenta registros que cumplen varios criterios al mismo tiempo.

=CONTAR.SI.CONJUNTO(rango_criterios1; criterio1; rango_criterios2; criterio2; ...)

Ejemplo BIOVAC

¿Cuántos lotes de Managua están aprobados?

=CONTAR.SI.CONJUNTO(D2:D101;"Managua";G2:G101;"Aprobado")

Aquí deben cumplirse las dos condiciones.

Otro ejemplo

¿Cuántos lotes de BIOVAC-20 en Managua tienen 25,000 unidades o más?

=CONTAR.SI.CONJUNTO(C2:C101;"BIOVAC-20";D2:D101;"Managua";F2:F101;">=25000")

Ejercicio guiado 5

- Cuenta los lotes aprobados de la planta Managua.
- Cuenta los lotes de BIOVAC-20 aprobados.
- Cuenta los lotes de León con menos de 20,000 unidades.

11. SUMAR.SI

SUMAR.SI suma valores que cumplen un criterio.

=SUMAR.SI(rango_criterio; criterio; rango_suma)

Ejemplo

¿Cuántas unidades se produjeron en Managua?

=SUMAR.SI(D2:D101;"Managua";F2:F101)

D contiene el criterio Planta y F contiene las Unidades que queremos sumar.

Ejercicio guiado 6

- Calcula las unidades producidas en Managua.
- Calcula las unidades producidas de BIOVAC-20.
- Calcula las unidades producidas por Ana López.

12. SUMAR.SI.CONJUNTO

SUMAR.SI.CONJUNTO permite sumar valores que cumplen varios criterios.

=SUMAR.SI.CONJUNTO(rango_suma; rango_criterio1; criterio1; rango_criterio2; criterio2; ...)

Ejemplo

¿Cuántas unidades aprobadas se produjeron en Managua?

=SUMAR.SI.CONJUNTO(F2:F101;D2:D101;"Managua";G2:G101;"Aprobado")

Ejemplo con tres criterios

¿Cuántas unidades de BIOVAC-20, aprobadas, se produjeron en Managua?

=SUMAR.SI.CONJUNTO(F2:F101;C2:C101;"BIOVAC-20";D2:D101;"Managua";G2:G101;"Aprobado")

CONTAR.SI.CONJUNTO vs SUMAR.SI.CONJUNTO

Necesidad	Función
¿Cuántos registros cumplen criterios?	CONTAR.SI.CONJUNTO
¿Cuánto suman los registros que cumplen criterios?	SUMAR.SI.CONJUNTO

13. Funciones de búsqueda y referencia

Ahora cambiaremos la pregunta. Ya no queremos solamente contar o sumar. Queremos encontrar información.

Ejemplo: el departamento administrativo recibe el código de lote BV-025 y necesita recuperar automáticamente la vacuna, planta, responsable, unidades y estado.

Antes de buscar, identifica tres elementos

15. ¿Qué valor conozco? → por ejemplo, el código de lote.

16. ¿Dónde está ese valor? → en la columna de códigos.

17. ¿Qué información quiero devolver? → por ejemplo, Responsable.

14. BUSCARV

BUSCARV busca un valor en la primera columna de una matriz y devuelve información de otra columna de la misma fila.

=BUSCARV(valor_buscado; matriz_buscar_en; indicador_columnas; [ordenado])

Ejemplo

Si F2 contiene BV-003 y la tabla A2:H101 contiene los lotes:

=BUSCARV(F2;A2:H101;5;FALSO)

Esta fórmula devuelve la información de la columna 5 del rango: Responsable.

Los cuatro argumentos

Argumento	Qué significa
valor_buscado	El dato que conocemos.
matriz_buscar_en	La tabla donde buscar.
indicador_columnas	Número de columna que queremos devolver dentro de la matriz.
FALSO	Coincidencia exacta.

Importante

En una búsqueda de códigos de lotes, productos o empleados, normalmente debemos usar FALSO para evitar coincidencias aproximadas.

Ejercicio guiado 7

Escribe un código de lote en una celda de consulta y utiliza BUSCARV para devolver Vacuna, Planta y Responsable.

15. BUSCARH

BUSCARH realiza una búsqueda horizontal. Busca en la primera fila de una tabla y devuelve información de una fila determinada.

Indicador	Enero	Febrero	Marzo	Abril
Producción	250000	280000	310000	295000
Lotes liberados	10	12	14	13

Pregunta: ¿cuál fue la producción de marzo?

=BUSCARH("Marzo";B1:E2;2;FALSO)

En este ejemplo, la fila 2 contiene los valores de producción.

¿Cuándo utilizarla?

Es útil cuando la estructura de la información está organizada horizontalmente. En bases de datos verticales, BUSCARV suele resultar más natural.

16. ÍNDICE

ÍNDICE devuelve el contenido de una posición dentro de un rango.

=ÍNDICE(matriz; núm_fila)

Ejemplo

Fila	Vacuna	Costo
1	BIOVAC-20	450
2	BIOVAC-FLU	380
3	BIOVAC-P	1250
4	BIOVAC-R	520

=ÍNDICE(C2:C5;3)

Resultado: 1250.

La función no necesita saber el nombre del producto. Recibe una posición y devuelve el valor de esa posición.

17. COINCIDIR

COINCIDIR busca un valor dentro de un rango y devuelve su posición.

=COINCIDIR(valor_buscado; matriz_buscada; tipo_de_coincidencia)

Ejemplo

=COINCIDIR("BIOVAC-P";B2:B5;0)

Resultado: 3.

El argumento 0 indica coincidencia exacta.

Idea clave

COINCIDIR encuentra la posición. ÍNDICE utiliza una posición para devolver el dato.

18. ÍNDICE + COINCIDIR

Al combinar ambas funciones podemos crear una búsqueda flexible.

Código	Vacuna	Costo
VAC001	BIOVAC-20	450
VAC002	BIOVAC-FLU	380
VAC003	BIOVAC-P	1250
VAC004	BIOVAC-R	520

Si F2 contiene VAC003 y queremos recuperar el costo:

=ÍNDICE(C2:C5;COINCIDIR(F2;A2:A5;0))

Cómo funciona

18. COINCIDIR busca VAC003 en A2:A5.

19. COINCIDIR devuelve la posición 3.

20. ÍNDICE toma la posición 3 en C2:C5.

21. ÍNDICE devuelve 1250.

Ventaja

Esta combinación permite separar el lugar donde buscamos del lugar de donde devolvemos el resultado. Es especialmente útil cuando la columna de retorno está a la izquierda o cuando se necesita mayor flexibilidad que BUSCARV.

Ejercicio guiado 8

Crea una consulta donde el alumno escriba un código de vacuna y Excel devuelva su nombre, presentación y costo usando ÍNDICE + COINCIDIR.

19. Errores frecuentes y cómo resolverlos

Error / situación	Causa probable	Qué revisar
#N/A	No se encontró el valor	Código, espacios, coincidencia exacta
#¡VALOR!	Tipo de dato o argumento incorrecto	Rangos y argumentos
#REF!	Referencia inválida	Rangos eliminados o modificados
Resultado inesperado	Criterio incorrecto	Texto, operadores y rangos
BUSCARV no encuentra	El valor no está en la primera columna	Estructura de la matriz
Conteo incorrecto	Rangos de distinto tamaño	Que todos los rangos correspondan

Espacios invisibles

Un texto puede parecer idéntico y no serlo si contiene espacios adicionales. ESPACIOS puede ayudar a limpiar textos.

=ESPACIOS(A2)

En el segundo encuentro profundizaremos en funciones de texto.

Consejo

Cuando una fórmula falla, no la reemplaces inmediatamente. Lee la fórmula, identifica cada argumento y comprueba uno por uno.

20. Buenas prácticas de trabajo

- Utiliza nombres de columnas claros.
- No mezcles datos con títulos, subtotales o filas vacías dentro de una base.
- Evita escribir manualmente resultados que Excel puede calcular.
- Utiliza coincidencia exacta para códigos y registros únicos.
- Comprueba los rangos antes de copiar fórmulas.
- Usa referencias absolutas cuando una tabla de consulta debe permanecer fija.
- Conserva una hoja de datos original sin modificar.
- Documenta fórmulas complejas si otras personas utilizarán el archivo.
- Evita combinar celdas dentro de bases de datos.
- Verifica los resultados con una segunda forma de cálculo cuando el dato sea importante.

Buenas prácticas para búsquedas

- El código buscado debe estar limpio y ser consistente.
- Los códigos deben ser únicos cuando se espera un solo resultado.
- Usa FALSO en BUSCARV/BUSCARH cuando necesites coincidencia exacta.
- Comprueba que el rango incluya todas las filas necesarias.
- Si una búsqueda no funciona, prueba primero con una coincidencia directa.

21. Ejercicios guiados

Ejercicio A — Clasificación de lotes

Utiliza la base BASE_LOTES. Crea una columna llamada Nivel de producción.

- $\geq 40,000$: Muy alta
- $\geq 30,000$: Alta
- $\geq 20,000$: Media
- $< 20,000$: Baja

Función principal: SI anidado.

Fórmula propuesta para comenzar: _____

Ejercicio B — Liberación

Crea una columna que indique "Listo" únicamente si Calidad = Aprobado Y Unidades $\geq 25,000$. En caso contrario, debe indicar "Revisar".

Función principal: SI + Y.

Fórmula: _____

Ejercicio C — Indicadores

Calcula:

- Número de lotes aprobados.
- Número de lotes de Managua.
- Número de lotes de Managua aprobados.
- Unidades producidas en Managua.
- Unidades aprobadas de BIOVAC-20 en Managua.

Ejercicio D — Consulta de lote

Crea una celda donde el usuario introduzca un código de lote. Recupera mediante BUSCARV la vacuna, planta, responsable y estado.

22. Ejercicios de práctica individual

Resuelve los siguientes ejercicios sin copiar las fórmulas de los ejemplos. Antes de escribir cada fórmula, explica qué debe hacer Excel.

Práctica 1 — Estado operativo

Un lote se clasifica como "Liberable" si Calidad = Aprobado y tiene al menos 30,000 unidades. Si no cumple, debe mostrar "No liberable".

Fórmula: _____

Práctica 2 — Prioridad

Un lote es "Prioritario" si tiene menos de 20,000 unidades O si su calidad está Pendiente.

Fórmula: _____

Práctica 3 — Conteo múltiple

¿Cuántos lotes de BIOVAC-FLU están aprobados en Managua?

Fórmula: _____

Práctica 4 — Producción

¿Cuántas unidades aprobadas de BIOVAC-P se produjeron en León?

Fórmula: _____

Práctica 5 — Búsqueda

Introduce un código de lote y devuelve el nombre del responsable mediante BUSCARV.

Fórmula: _____

Práctica 6 — ÍNDICE + COINCIDIR

Introduce un código de vacuna y devuelve su costo unitario utilizando ÍNDICE + COINCIDIR.

Fórmula: _____

23. Reto integrador

Construye un pequeño módulo de consulta para el área administrativa de Laboratorios BIOVAC.

Parte A — Indicadores

Calcula automáticamente:

- Total de lotes
- Lotes aprobados
- Lotes pendientes
- Lotes rechazados
- Unidades totales producidas
- Unidades aprobadas

Parte B — Consulta

Crea un campo llamado Código de lote. Al escribir un código válido, deben aparecer:

- Vacuna
- Fecha
- Planta
- Responsable
- Unidades
- Calidad
- Estado

Parte C — Clasificación

Agrega una clasificación automática de nivel de producción:

- $\geq 40,000$ → Muy alta
- $\geq 30,000$ → Alta
- $\geq 20,000$ → Media
- $< 20,000$ → Baja

Parte D — Validación

Comprueba tu resultado seleccionando al menos tres códigos diferentes y verificando manualmente la información.

Criterios de evaluación

Criterio	Ponderación
Uso correcto de SI / SI anidado	20%
Uso de Y / O	15%
Conteo con criterios	15%
Suma con criterios	15%
Búsqueda con BUSCARV	15%
ÍNDICE + COINCIDIR	15%
Orden y claridad del archivo	5%

24. Autoevaluación

Marca mentalmente o en tu hoja de trabajo el nivel que consideras alcanzado.

Competencia	Necesito practicar	En proceso	Lo domino
Construyo una fórmula SI	☐	☐	☐
Construyo SI anidado	☐	☐	☐
Combino SI con Y/O	☐	☐	☐
Uso CONTAR.SI	☐	☐	☐
Uso CONTAR.SI.CONJUNTO	☐	☐	☐
Uso SUMAR.SI	☐	☐	☐
Uso SUMAR.SI.CONJUNTO	☐	☐	☐
Uso BUSCARV	☐	☐	☐
Uso BUSCARH	☐	☐	☐
Comprendo ÍNDICE	☐	☐	☐
Comprendo COINCIDIR	☐	☐	☐
Combino ÍNDICE + COINCIDIR	☐	☐	☐

Preguntas de reflexión

22. ¿Qué función utilizarías para saber cuántos registros cumplen un criterio?
23. ¿Qué función utilizarías para sumar valores con dos o más condiciones?
24. ¿Cuál es la diferencia entre Y y O?
25. ¿Por qué utilizar FALSO en una búsqueda de códigos?
26. ¿Qué función encuentra una posición: ÍNDICE o COINCIDIR?
27. ¿Qué función devuelve el dato de una posición: ÍNDICE o COINCIDIR?

25. Hoja de fórmulas de consulta rápida

Función	Sintaxis básica	Uso
SI	=SI(condición;verdadero;falso)	Tomar decisiones
Y	=Y(condición1;condición2)	Todas las condiciones
O	=O(condición1;condición2)	Al menos una condición
CONTAR.SI	=CONTAR.SI(rango;criterio)	Contar con un criterio
CONTAR.SI.CONJUNTO	=CONTAR.SI.CONJUNTO(rango1;criterio1;...)	Contar con varios criterios
SUMAR.SI	=SUMAR.SI(rango;criterio;suma)	Sumar con un criterio
SUMAR.SI.CONJUNTO	=SUMAR.SI.CONJUNTO(suma;rango1;criterio1;...)	Sumar con varios criterios
BUSCARV	=BUSCARV(valor;matriz;columna;FALSO)	Búsqueda vertical
BUSCARH	=BUSCARH(valor;matriz;fila;FALSO)	Búsqueda horizontal
ÍNDICE	=ÍNDICE(rango;posición)	Devolver valor por posición
COINCIDIR	=COINCIDIR(valor;rango;0)	Encontrar posición
ÍNDICE+COINCIDIR	=ÍNDICE(rango_retorno;COINCIDIR(valor;rango_búsqueda;0))	Búsqueda flexible

Operadores útiles

Operador	Significado	Ejemplo
=	Igual	=A2="Managua"
>	Mayor que	>25000
<	Menor que	<20000
>=	Mayor o igual	>=25000
<=	Menor o igual	<=25000
<>	Distinto	<>"Aprobado"

26. Glosario

Término	Definición
Criterio	Condición utilizada para seleccionar o evaluar información.
Rango	Conjunto de celdas utilizado por una fórmula.
Matriz	Tabla o rango donde se realiza una búsqueda.
Coincidencia exacta	La búsqueda debe encontrar exactamente el valor indicado.
Referencia relativa	Referencia que cambia al copiar una fórmula.
Referencia absoluta	Referencia que permanece fija al copiar una fórmula.
Base de datos	Conjunto organizado de registros y campos.
Registro	Una fila que representa un elemento de información.
Campo	Una columna que representa una característica del registro.
Función	Fórmula predefinida que realiza una operación.

Cierre del encuentro

En este encuentro aprendiste a utilizar Excel para tomar decisiones, contar y sumar información bajo condiciones y realizar búsquedas dentro de una base de datos. El siguiente paso será trabajar con bases de datos más grandes, filtros, subtotales, consolidación y funciones de texto, fecha y hora.

Antes de terminar, guarda tu archivo de práctica y conserva una copia del reto integrador para utilizarlo como referencia durante el curso.